
Application (A1; "WordPerfect"; UK)

Global (CommandFile)
CommandFile := GetValue ("Command Files"; "MYOB"; "swdrvs.ini")

TFile := GetValue ("Template"; "TemplateFName"; CommandFile)
TemplateSelect(TFile)

// ************** GET THE TAGS ***********************

NumPrompts:=0
OnNotFound(NoMore)
OnError(NoMore)

WRITEDLL ("Settings"; "Run Flag" ; "1" ;CommandFile)

PosDocVeryTop

Label (SearchLoop)

// Main search loop. Searches for tags with  .. 

SearchString (StrgToLookFor: "[[")
MatchPositionBefore ()
SearchNext (SearchMode: Extended!)
SelectMode (State: On!)
SearchString (StrgToLookFor: "]]")
MatchExtendSelection ()
SearchNext (SearchMode: Regular!)

TemplateField := ?SelectedText
NumPrompts:=NumPrompts+1

SelectMode(Off!)

TPrompt := NumStr(NumPrompts)
WRITEDLL ("Tags"; "num" ;TPrompt ; CommandFile)
WRITEDLL ("Tags"; TPrompt ; TemplateField ; CommandFile)

Go (SearchLoop)

Label(NoMore)
While (?SubStructure)

SubStructureExit



Type("x")
DeleteCharPrevious

EndWhile

WRITEDLL ("Settings"; "Run Flag" ; "0" ;CommandFile)

Quit

Procedure WRITEDLL (Section;Tag;StrValue;IniFileName)

If (?MajorVersion>6)
DLLLoad(Kernel;"Kernel32.DLL")

else
DLLLoad(Kernel;"Kernel.DLL")

endif

DLLCall(Kernel;"WritePrivateProfileString";itemnum:WORD;
{section;Tag;StrValue;IniFileName})
DLLFree(Kernel)
EndProc

//*************************************************************************

Function GetValue (section;Tag;IniFile)

If (?MajorVersion>6)
DLLLoad(Kernel;"Kernel32.DLL")

else
DLLLoad(Kernel;"Kernel.DLL")

endif

ReturnValue:=""
DLLCall(Kernel;"GetPrivateProfileString";itemnum:WORD;
{section;Tag;"";Address(ReturnValue);Loword(127);IniFile})
DLLFree(Kernel)

Return(ReturnValue)

EndFunc

//**********************************************************


